Using P-Elements

Introduction

This tutorial was completed using ANSYS 7.0. This tutorial outlines the steps necessary for solving a model meshed with p-elements. The p-method manipulates the polynomial level (p-level) of the finite element shape functions which are used to approximate the real solution. Thus, rather than increasing mesh density, the p-level can be increased to give a similar result. By keeping mesh density rather coarse, computational time can be kept to a minimum. This is the greatest advantage of using p-elements over h-elements.

A uniform load will be applied to the right hand side of the geometry shown below. The specimen was modeled as steel with a modulus of elasticity of 200 GPa.



Preprocessing: Defining the Problem

1. **Give example a Title**

   Utility Menu > File > Change Title ...
   `/title, P-Method Meshing`

2. **Activate the p-Method Solution Options**

   ANSYS Main Menu > Preferences
   `/PMETH,ON`

   Select **p-Method Struct.** as shown below



3. **Open preprocessor menu**
   ANSYS Main Menu > Preprocessor
   `/PREP7`
4. **Define Keypoints**
   Preprocessor > Modeling > Create > Keypoints > In Active CS...
   `K,#,x,y,z`
   We are going to define 12 keypoints for this geometry as given in the following table:

Keypoint	Coordinates (x,y,z)
1	(0,0)
2	(0,100)
3	(20,100)
4	(45,52)
5	(55,52)
6	(80,100)
7	(100,100)
8	(100,0)
9	(80,0)
10	(55,48)
11	(45,48)
12	(20,0)

5. **Create Area**
   Preprocessor > Modeling > Create > Areas > Arbitrary > Through KPs
   `A,1,2,3,4,5,6,7,8,9,10,11,12`
   Click each of the keypoints in numerical order to create the area shown below.



6. **Define the Type of Element**
   Preprocessor > Element Type > Add/Edit/Delete...
   For this problem we will use the PLANE145 (p-Elements 2D Quad) element. This element has eight nodes with 2 degrees of freedom each (translation along the X and Y axes). It can support a polynomial with maximum order of eight.

After clicking OK to select the element, click **Options...** to open the keyoptions window, shown below. Choose **Plane stress + TK** for Analysis Type.



Keyopts 1 and 2 can be used to set the starting and maximum p-level for this element type. For now we will leave them as default.

Other types of p-elements exist in the ANSYS library. These include Solid127 and Solid128 which have electrostatic DOF's, and Plane145, Plane146, Solid147, Solid148 and Shell150 which have structural DOF's. For more information on these elements, go to the Element Library in the help file.

7. **Define Real Constants**

   Preprocessor > Real Constants... > Add...

   In the 'Real Constants for PLANE145' window, enter the following geometric properties:

   i. Thickness THK: 10

   This defines an element with a thickness of 10 mm.

8. **Define Element Material Properties**

   Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic

   In the window that appears, enter the following geometric properties for steel:

   i. Young's modulus EX: 200000

   ii. Poisson's Ratio PRXY: 0.3

9. **Define Mesh Size**

   Preprocessor > Meshing > Size Cntrls > ManualSize > Areas > All Areas...

   For this example we will use an element edge length of 5mm.

10. **Mesh the frame**

    Preprocessor > Meshing > Mesh > Areas > Free > click 'Pick All'

Solution Phase: Assigning Loads and Solving

1. **Define Analysis Type**

   Solution > Analysis Type > New Analysis > Static

   `ANTYPE,0`

2. **Set Solution Controls**

   Solution > Analysis Type > Sol'n Controls

   The following window will pop up.



A) Set `Time at end of loadstep` to 1 and `Automatic time stepping` to ON

B) Set `Number of substeps` to 20, `Max no. of substeps` to 100, `Min no. of substeps` to 20.

C) Set the `Frequency` to Write every substep

3. **Apply Constraints**

   Solution > Define Loads > Apply > Structural > Displacement > On Lines

   Fix the left side of the area (ie all DOF constrained)

4. **Apply Loads**

   Solution > Define Loads > Apply > Pressure > On Lines

   Apply a pressure of -100 N/mm^2

   The applied loads and constraints should now appear as shown in the figure below.



5. **Solve the System**

   Solution > Solve > Current LS

   `SOLVE`

Postprocessing: Viewing the Results

1. **Read in the Last Data Set**

   General Postproc > Read Results > Last Set

2. **Plot Equivalent Stress**

   General Postproc > Plot Results > Contour Plot > Element Solu

   In the window that pops up, select Stress > von Mises SEQV



   The following stress distribution should appear.



3. **Plot p-Levels**

   General Postproc > Plot Results > p-Method > p-Levels

   The following distribution should appear.



Note how the order of the polynomial increased in the area with the greatest range in stress. This allowed the elements to more accurately model the stress distribution through that area. For more complex geometries, these orders may go as high as 8. As a comparison, a plot of the stress distribution for a normal h-element (PLANE2) model using the same mesh, and one with a mesh 5 times finer are shown below.





As one can see from the two plots, the mesh density had to be increased by 5 times to get the accuracy that the p-elements delivered. This is the benefit of using p-elements. You can use a mesh that is relatively coarse, thus computational time will be low, and still get reasonable results. However, care should be taken using p-elements as they can sometimes give poor results or take a long time to converge.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to **'File > Read input from...'** and select the file.

```
finish
/clear

/title, P-Method Meshing
/pmeth,on                       ! Initialize p-method in ANSYS

/prep7                          ! Enter preprocessor

k,1,0,0                         ! Keypoints defining geometry
k,2,0,100
k,3,20,100
k,4,45,52
k,5,55,52
k,6,80,100
k,7,100,100
k,8,100,0
k,9,80,0
k,10,55,48
k,11,45,48
k,12,20,0

a,1,2,3,4,5,6,7,8,9,10,11,12   ! Create area from keypoints

et,1,plane145                   ! Element type
keyopt,1,3,3                    ! Plane stress with thickness option
```

```
r,1,10                          ! Real constant - thickness
mp,ex,1,200000                  ! Young's modulus
mp,prxy,1,0.3                   ! Poisson's ratio

esize,5                         ! Element size
amesh,all                       ! Mesh area

finish
/solu                           ! Enter solution phase

antype,0                        ! Static analysis
nsubst,20,100,20                ! Number of substeps
outres,all,all                  ! Output data for all substeps
time,1                          ! Time at end = 1

lsel,s,loc,x,0                  ! Line select at x=0
dl,all,,all                     ! Constrain the line, all DOF's
lsel,all                        ! Re-select all lines

lsel,s,loc,x,100                ! Line select at x=100
sfl,all,pres,-100               ! Apply a pressure
lsel,all                        ! Re-select all lines

solve
finish

/post1                          ! Enter postprocessor
set,last                        ! Select last set of data
plesol,s,eqv                    ! Plot the equivalent stress
```

Reference

http://www.mece.ualberta.ca/tutorials/ansys